# This file is part of BlackArch Linux ( https://www.blackarch.org/ ).
# See COPYING for license details.

pkgname=python-requests-html
_pkgname=requests-html
pkgver=462.075ac16
pkgrel=4
pkgdesc='Pythonic HTML Parsing for Humans.'
arch=('any')
url='https://github.com/psf/requests-html'
license=('MIT')
depends=('python' 'python-requests' 'python-pyquery' 'python-fake-useragent'
         'python-parse' 'python-beautifulsoup4' 'python-w3lib'
         'python-pyppeteer')
makedepends=('git' 'python-setuptools')
source=("git+https://github.com/psf/$_pkgname.git")
sha512sums=('SKIP')

pkgver() {
  cd $_pkgname

  echo $(git rev-list --count HEAD).$(git rev-parse --short HEAD)
}

build() {
  cd $_pkgname

  python setup.py build
}

package() {
  cd $_pkgname

  python setup.py install --prefix=/usr --root="$pkgdir" -O1 --skip-build
}

